BenderOrszag Ex1p241 03-31-16

N. T. Gladd

Initialization: Be sure the files *NTGStylesheet2.nb* and *NTGUtilityFunctions.m* is are in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing “shift” + “enter”. Respond “Yes” in response to the query to evaluate initialization cells.

In[7]:=
```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
SetOptions[EvaluationNotebook[], (* load the StyleSheet *)
 StyleDefinitions → Get["NTGStylesheet2.nb"]];
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

Cleaned up version of work originally performed in *BenderOrszag Ex1p241 09-12-15*

Purpose

Section 6.6 of Bender and Orszag deals with evaluation of an integral using the method of steepest descent. Given an integral of the form

$$\mathcal{I}(k) = \int_C \mathrm{d}z\, h(z)\, e^{k\,\rho(z)}$$

the idea of steepest descent is deform the contour so that the integral can be approximated by

$$\mathcal{I}(k) = \int_C \mathrm{d}z\, h(z)\, e^{k\,(\rho_R(z)+i\,\rho_I(z))} \simeq e^{i\,k\,\rho_I} \int_{C_{\text{deformed}}} \mathrm{d}z\, h(z)\, e^{k\ \rho_R(z)}$$

In words, a contour is sought along which the imaginary part of $\rho(z)$ is approximately constant. Under such circumstances the integrand is localized and as well as not highly oscillatory at large k.

Example 1, p241 of Bender and Orszag considers valuation of

$$\mathcal{I}(k) = \int_0^1 \ln(z)\, e^{i\,k\,z}\, dz$$

After working through this calculation with pen and paper, I perform the calculation using *Mathematica* and provide some visualizations.

I begin with some visualizations of the function over which the integration is performed.
Note the branch cut associated with log and the poles associated with exp[i k z] for $z_{\text{Im}} < 0$

Out[9]=



$\mathcal{I}$(k) is evaluated by contour integration. The contour C (0,0) → (1, 0) is deformed into C_1 (0,0) → (0, i Y) + C_2 (0, i Y) → (1, i Y) + C_3 (1, i Y) → (0, i Y). Note that C_1 and C_3 are along line of constant phase {x, y} = r $e^{i\pi/2}$



It is instructive to view a 3-D representation of the integration path



Analysis

I develop analytical approximations for the integrations along contours C_1, C_2, C_3.

I C_1

Along contour C_1 the integrand is

In[13]:=
```
w1[1] =
 Log[z] Exp[I k z] dz /. z → 0 + I y /. dz → I dy /. dy → 1 // PowerExpand // Expand
```

Out[13]=
$$-\frac{1}{2} e^{-k y} \pi + i\, e^{-k y} \text{Log}[y]$$

I use a constructed function *Int* (rather than Mathematica's *Integrate*) to temporarily suppress evaluation of the integral while manipulations are performed. After manipulations have been completed, evaluation can be triggered by setting Int → Integrate. There is a slight complication in that Integrate is designed for integration on the real line and the concern here is with contour integrals. Note that I explicitly introduce a differential form dz for the variable of integration z, and define it appropriately for the contour being handled. This insures that any complex phase factor has been properly incorporated in the calculation. For example, a contour along the real axis would have dz → dx and a contour along the imaginary axis would have dz → I dy. Finally, since Integrate does not require explicit mention of the differential dx or dy, I set them to unity.

In[14]:=
```
w1[2] = Int[w1[1], {y, 0, Y}] //. IntRules
```

Out[14]=
$$-\frac{1}{2} \pi \text{Int}\left[e^{-k y}, \{y, 0, Y\}\right] + i \text{Int}\left[e^{-k y} \text{Log}[y], \{y, 0, Y\}\right]$$

where simplifying rules IntRules are defined below.

Take the limit Y → ∞

In[15]:=
```
w1[3] = w1[2] /. Y → ∞
```

Out[15]=
$$-\frac{1}{2} \pi \text{Int}\left[e^{-k y}, \{y, 0, \infty\}\right] + i \text{Int}\left[e^{-k y} \text{Log}[y], \{y, 0, \infty\}\right]$$

In[16]:=
```
w1[4] =
 (w1[3][[1]] /. Int → Integrate // Simplify[#, {k ∈ Reals, k > 0}] &) + w1[3][[2]]
```

Out[16]=
$$-\frac{\pi}{2 k} + i \text{Int}\left[e^{-k y} \text{Log}[y], \{y, 0, \infty\}\right]$$

Evaluating the integral

In[17]:=
```
w1["final"] =
 w1[4] /. Int → Integrate // Simplify[#, {k ∈ Reals, k > 0}] & // Expand
```

Out[17]=
$$-\frac{i\, \text{EulerGamma}}{k} - \frac{\pi}{2 k} - \frac{i \text{Log}[k]}{k}$$

In[18]:=
```
QuickCheck[NIntegrate[ w1[1] /. k → 3, {y, 0, ∞}], w1["final"] /. k → 3]
```

Out[18]=

Check

numerical	approximate	|difference|
−0.523599 − 0.558609 ⅈ	−0.523599 − 0.558609 ⅈ	2.40461×10^{-12}

2 C_2

Along this contour the integrand is

In[19]:=
```
w2[1] = Log[z] Exp[I k z] dz /. z → x + I Y  /. dz → dx /. dx → 1 // ExpandAll
```

Out[19]=
$$e^{i k x - k Y} \text{Log}[x + i Y]$$

As Y → ∞, the integrand tends to 0 and the contribution from this integal is 0

In[20]:=
```
w2[2] = Limit[w2[1], Y → ∞, Assumptions → {k > 0}]
```

Out[20]=
```
0
```

In[21]:=
```
w2["final"] = w2[2]
```

Out[21]=
```
0
```

3 C_3

In[22]:=
```
w3[1] = Log[z] Exp[I k z] dz /. z → 1 + I y  /. dz → I dy /. dy → 1 // ExpandAll
```

Out[22]=
$$i\, e^{i k - k y} \text{Log}[1 + i y]$$

This integral is approximated by expanding the logarithm and performing term by term integration

In[23]:=
```
w3[2] = Series[ Log[1 + ⅈ y], {y, 0, 5}] // Normal
```

Out[23]=
$$i y + \frac{y^2}{2} - \frac{i y^3}{3} - \frac{y^4}{4} + \frac{i y^5}{5}$$

In[24]:=
```
w3[3] = w3[1] /. Log[1 + I y] → w3[2]
```

Out[24]=
$$i\, e^{i k - k y} \left(i y + \frac{y^2}{2} - \frac{i y^3}{3} - \frac{y^4}{4} + \frac{i y^5}{5} \right)$$

In[25]:=
```
w3[4] = Int[w3[3] // Expand, {y, ∞, 0}] //. IntRules
```

Out[25]=
$$-\text{Int}\left[e^{i\,k-k\,y}\,y,\ \{y,\ \infty,\ 0\}\right]+\frac{1}{2}\,i\,\text{Int}\left[e^{i\,k-k\,y}\,y^2,\ \{y,\ \infty,\ 0\}\right]+\frac{1}{3}\,\text{Int}\left[e^{i\,k-k\,y}\,y^3,\ \{y,\ \infty,\ 0\}\right]-\frac{1}{4}\,i\,\text{Int}\left[e^{i\,k-k\,y}\,y^4,\ \{y,\ \infty,\ 0\}\right]-\frac{1}{5}\,\text{Int}\left[e^{i\,k-k\,y}\,y^5,\ \{y,\ \infty,\ 0\}\right]$$

In[26]:=
```
w3[5] = w3[4] /. Int → Integrate // Simplify[#, {k ∈ Reals, k > 0}] & // Expand
```

Out[26]=
$$\frac{24\,e^{i\,k}}{k^6}+\frac{6\,i\,e^{i\,k}}{k^5}-\frac{2\,e^{i\,k}}{k^4}-\frac{i\,e^{i\,k}}{k^3}+\frac{e^{i\,k}}{k^2}$$

In[27]:=
```
QuickCheck[NIntegrate[w3[1] /. k → 5, {y, ∞, 0}], w3[5] /. k → 5 ]
```

Out[27]=

Check

numerical	approximate	|difference|
0.00417302 – 0.0380059 *i*	0.00504421 – 0.038486 *i*	0.000994696

Simplify so that a pattern can be detected

In[28]:=
```
w3[6] = (# term) & /@ w3[5]  /. term → k^2 / Exp[I k]
```

Out[28]=
$$1+\frac{24}{k^4}+\frac{6\,i}{k^3}-\frac{2}{k^2}-\frac{i}{k}$$

I can see the general pattern

In[29]:=
```
Table[(-I)^n n!/k^n, {n, 0, 4}]
```

Out[29]=
$$\left\{1,\ -\frac{i}{k},\ -\frac{2}{k^2},\ \frac{6\,i}{k^3},\ \frac{24}{k^4}\right\}$$

and check that this is correct

In[30]:=
```
w3[7] = (1/term Sum[(-I)^n n!/k^n, {n, 0, 4}] /. term → k^2 / Exp[I k]) - w3[5] // ExpandAll
```

Out[30]=
```
0
```

In[31]:=
```
w3["final"] = 1/term sum[(-I)^n n!/k^n, {n, 0, ∾}] /. term → k^2 / Exp[I k]
```

Out[31]=
$$\frac{e^{i\,k}\,\text{sum}\left[(-i)^n\,k^{-n}\,n!,\ \{n,\ 0,\ \infty\}\right]}{k^2}$$

In[32]:=
```
QuickCheck[NIntegrate[w3[1] /. k → 5, {y, ∞, 0}] /. k → 5,
 w3["final"] /. k → 5 /. 𝒩 → 5 /. sum → Sum]
```

Out[32]=

Check

numerical	approximate	|difference|
0.00417302 – 0.0380059 *i*	0.00357131 – 0.0389217 *i*	0.00109574

4 Combining the contributions from the three contours and checking against numerical integration

In[33]:=
```
w4[1] = w1["final"] + w2["final"] + w3["final"]
```

Out[33]=

$$-\frac{i\,\text{EulerGamma}}{k}-\frac{\pi}{2\,k}-\frac{i\,\text{Log}[k]}{k}+\frac{e^{i\,k}\,\text{sum}\left[\left(-i\right)^{n}k^{-n}\,n!,\,\{n,\,0,\,\mathcal{N}\}\right]}{k^{2}}$$

In[34]:=
```
Module[{Y = 10, 𝒩 = 3, kMin = 4, kMax = 20,
  numericalValues, points, ℐnumerical, ℐanalytical, lab},
 ℐnumerical[k_] :=
  NIntegrate[Log[z] Exp[I k z] dz /. z → I y /. dz → I dy /. dy → 1, {y, 0, Y}] +
   NIntegrate[Log[z] Exp[I k z] /. z → x + I Y /. dz → dx /. dx → 1, {x, 0, 1}] +
   NIntegrate[Log[z] Exp[I k z] /. z → 1 + I Y /. dz → I dy /. dy → 1, {y, Y, 0}];
 Ianalytical[k_] := -(I EulerGamma)/k - π/(2 k) - (I Log[k])/k + 1/k^2 E^(I k) Sum[(-I)^n k^-n n!, {n, 0, 𝒩}];
 numericalValues =
  Table[{k, Re[ℐnumerical[k]], Im[ℐnumerical[k]]}, {k, kMin, kMax, 1}];
 lab =
  Stl["Asymptotic approximation of ℐ(k)\ncircles come from numerical integration"];
 Plot[{Re[Ianalytical[k]], Im[Ianalytical[k]]}, {k, kMin, kMax},
  PlotStyle → {Black, Directive[Black, Dashed]},
  AxesLabel → {Stl["k"], Stl["ℐ(k)"]},
  Epilog → {OC[#, Black] & /@ numericalValues[[All, {1, 2}]],
    OC[#, Black] & /@ numericalValues[[All, {1, 3}]]}, PlotLabel → lab]]
```

Out[34]=



Functions

Int is a convenience function for manipulating integral. *IntRules* are some rewrite rules for simplifying *Int[...]*.

In[3]:=
```
Clear[IntRules];
IntRules =
  (* Make Int Associative*){Int[a_ + b_, {var_, lowerLim_, upperLim_}] →
    Int[a, {var, lowerLim, upperLim}] + Int[b, {var, lowerLim, upperLim}],
   (* Make Int transparent to terms free of the integration variable *)
   Int[a_. b_, {var_, lowerLim_, upperLim_}] /; FreeQ[a, var] →
    a Int[b, {var, lowerLim, upperLim}]};
```

QuickCheck is a convenience function for checking that an error has not occurred during a sequence of

intermediate calculations.

In[5]:=
```
Clear[QuickCheck];
QuickCheck[numericalIntegral_, approximateIntegral_] :=
 Module[{num, approx, info, diff},
  num = numericalIntegral;
  approx = approximateIntegral // N;
  diff = Abs[num - approx];
  info = LGrid[
    {{"numerical", "approximate", "|difference|"}, {num, approx, diff}}, "Check"]]
```

Graphics

In[10]:=
```
Module[{k = 2, imageSize = 400, lab, C, g, F, DirectiveList, CArrow},
 F[z_, k_] := Log[z] Exp[I k z];
 DirectiveList[plotLabel_, {xLabel_, yLabel_, zLabel_}] :=
  Sequence[PlotLabel → plotLabel, AxesLabel → {xLabel, yLabel, zLabel},
   ImageSize → imageSize, MeshFunctions -> {#3 &}, Mesh → 5];
 CArrow[{p1_, p2_}, {lab_, offset_}] :=
  {PointSize[0.015], Point[p1], Point[p2], Arrow[{p1, (p1 + p2) / 2}],
   Line[{(p1 + p2) / 2, p2}], Style[Text[lab, offset + p1 + (p2 - p1) / 2], Small]};

 lab =
  Stl@StringForm["k = ``  F(z, k) = ``", k, TraditionalForm[Log[z] Exp[I k z]]];
 C = {BLACK, CArrow[{{0, 0}, {1, 0}}, {Stl["C"], {0, 0.25}}]};
 g[1] = Plot3D[Re[F[x + I y, k]], {x, -3, 3},
   {y, -3, 3}, Evaluate[DirectiveList[lab, {"x", "y", "Re[F]"}]]];
 g[2] = ContourPlot[Re[F[x + I y, k]], {x, -3, 3}, {y, -3, 3},
   AxesLabel → {"x", "y", ""}, ImageSize → 300, Epilog → C];
 g[3] = Plot3D[Im[F[x + I y, k]], {x, -3, 3}, {y, -3, 3},
    Evaluate[DirectiveList["", {"x", "y", "Im[F]"}]]];
 g[4] = ContourPlot[Im[F[x + I y, k]], {x, -3, 3}, {y, -3, 3},
   AxesLabel → {"x", "y", ""}, ImageSize → 300, Epilog → C];
 Grid[{{g[1], g[2]}, {g[3], g[4]}}]]
```

Out[10]=



In[11]:=

```
Module[{k = 2, Y = 2, image = 300, g, C, C1, C2, C3, F, CArrow},
 F[z_, k_] := Log[z] Exp[I k z];
 CArrow[{p1_, p2_}, {lab_, offset_}] :=
  {PointSize[0.015], Point[p1], Point[p2], Arrow[{p1, (p1 + p2) / 2}],
   Line[{(p1 + p2) / 2, p2}], Style[Text[lab, offset + p1 + (p2 - p1) / 2], Small]};

 C = {BLACK, CArrow[{{0, 0}, {1, 0}}, {"C", {0, 0.25}}]};
 C1 = {BLUE, CArrow[{{0, 0}, {0, Y}}, {"C1", {-0.05, 0}}]};
 C2 = {BLUE, CArrow[{{0, Y}, {1, Y}}, {"C2", {0, 0.05}}]};
 C3 = {BLUE, CArrow[{{1, Y}, {1, 0}}, {"C3", {0.05, 0}}]};
 g[1] = ContourPlot[Re[F[x + I y, k]], {x, -1, 2}, {y, -1, 3},
   AxesLabel → {"x", "y", ""}, ImageSize → image,
   Epilog → {C, C1, C2, C3}, FrameLabel → Stl["Re[F]"]];
 g[2] = ContourPlot[Im[F[x + I y, k]], {x, -1, 2}, {y, -1, 3},
   AxesLabel → {"x", "y", ""}, ImageSize → image,
   Epilog → {C, C1, C2, C3}, FrameLabel → Stl["Re[F]"]];
 Grid[{{g[1], g[2]}}]]
```

Out[11]=



In[12]:=

```
Module[{k = 2, Y = 2, imageSize = 300, lab, g, C, C1, C2,
  C3, gCurve, curve, surface, F, PointOnContour, ContourPath3D},
 F[z_, k_] := Log[z] Exp[I k z];
 PointOnContour[x_, y_, k_] :=
  {x, y, Abs[F[x + I y, k]] + 0.01};
 ContourPath3D[points_, text_] :=
  Module[{n = Length@points, nStart, nFin, nMid, arrow, txt},
   {nStart, nFin} = {Round[0.2 n], Round[0.25 n]};
   nMid = Round[0.5 n];
   arrow = Arrow[{points [[nStart]], points [[nFin]]}];
   txt = Text[text, points [[nMid]]];
   {Line[points], arrow, txt}] ;

 lab = Stl[StringForm["|F(z, ``)|", k]];
 C = ContourPath3D[
   Table[PointOnContour[x, 0, k], {x, 0.01, 1, 0.01}], Style["C", 16, Bold, Black]];
 C1 = ContourPath3D[Table[PointOnContour[0, y, k], {y, 0.01, 2, 0.01}],
   Style["C1", 16, Bold, Black]];
 C2 = ContourPath3D[Table[PointOnContour[x, 2, k], {x, 0.01, 1, 0.01}],
   Style["C2", 16, Bold, Black]];
 C3 = ContourPath3D[Table[PointOnContour[1, y, k], {y, 2, 0.01, -0.01}],
   Style["C3", 16, Bold, Black]];
 gCurve = Graphics3D[{{RED, C}, {BLUE, C1, C2, C3}},
   PlotRange → {{-0.5, 2}, {-0.5, 3}, {0, 5}}, Axes → Automatic,
   AxesLabel → {Stl["x"], Stl["y"], lab}, ImageSize → imageSize, Boxed → False];
 surface = Plot3D[Abs[F[x + I y, k]], {x, -1, 2}, {y, -1, 3},
   ImageSize → imageSize, MeshFunctions -> {#3 &}, Mesh → 10, Boxed → False];
 Show[{gCurve, surface}]]
```

Out[12]=